

Part 1: SAS (Sgplot プロシジャ) や可視化の導入

0. 無料版 SAS の紹介

もし SAS の実行環境をお持ちでない場合、学習目的に限り、クラウド上で無償版 SAS が利用できます。基本的には本資料と同様のコードが実行でき、同様の結果を得ることができます。ただし、利用データをクラウドにアップロード、ライブラリ指定などの手順で、コードや機能に製品版と差異があるので、詳しくは SAS 社、公式情報を参照ください。

[SAS® ONDEMAND FOR ACADEMICS]

https://www.sas.com/ja_jp/software/on-demand-for-academics.html#8239eb92-b4a3-4ef9-9306-f708a14164e9

1. Dataset の準備

[What If の NHEFS data]

https://cdn1.sph.harvard.edu/wp-content/uploads/sites/1268/2012/10/nhefs_sas.zip

ダウンロードし、解凍した nhefs.sas7bdat を、参照可能な適当な場所に保存します。そして

```
libname lib1 “保存したパス”;
```

これにより、ライブラリ lib1 の nhefs データセットとして認識可能。

libname ステートメントで明示的に命名されたロケーションを SAS では永久ライブラリといい、なにも指定せずにデータ加工する一時的な領域を Work ライブラリと呼びます。

2. 作図の基本文法

一番基本的なコマンドは、sgplot プロシジャを使った次のようなものになります

```
proc sgplot data = lib1.nhefs;  
  scatter x=sbp y=dbp;  
run;
```



```
proc sgplot data = [ライブラリ名.データセット名];
  [プロットステートメント] ---プロットステートメントごとの必須な指定 /オプション ;
run;
```

といった形が基本の構文スタイルになります。

先の例では[プロットステートメント] として scatter を使い、散布図を作成しました
scatter を指定した場合の必須指定は x = と y =になります。

必須要素以外のオプションが必要な場合はスラッシュ[/]を挟んでオプションを指定します。


指定できるプロットステートメントと、必須指定およびオプション指定は、オンラインのリファレンスを参照します

https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatproc/p073b197jzadkmn15lhq58yiy2uh.htm

[例 1] : 推移図 [series ステートメント]


```
proc sgplot data=sashelp.pricedata;
  series x = date y = sale / group=product;
run;
```

/*sashelp.ライブラリは備え付けのライブラリで、サンプルデータセットも入っている*/




[例 2] : ヒストグラム [hisotgram ステートメント]

```
proc sgplot data = lib1.nhefs;
  histogram sbp;
run;
```



[例 3] : 箱ひげ図 [縦向きの場合は vbox, 横向きの場合は hbox]


```
proc sgplot data = lib1.nhefs;
  vbox ht / category=sex;
run;
```



[例 4]: グラフの重ね合わせ

基本的に `sgplot;~run;`の間に複数のプロットステートメントを挿入することができ、複数のグラフを重ねることができます

```
proc sgplot data = lib1.nhefs;
  scatter x=sbp y=dbp;
  reg x=sbp y=dbp;
run;
```



↑ `scatter`[散布図]に `reg`[回帰直線]を重ねている。

3. グラフ体裁の調整方法

3.1 プロットごとの属性とグループごとの属性

プロットごとの体裁の場合、基本的にはオプションで


`XXXXattrs = (変えたい属性 = 値 , 変えたい属性 = 値 ...)`と記述します

`XXXX`の部分は体裁を変えたい要素であり、マーカーの色や形を変えたければ

`markerattrs`になり、線の色や線種を変えたければ`lineattrs`になります。

散布図のマーカーの形を塗りつぶし四角にし、色を黒、サイズを5ptとして、回帰直線の色を赤、太さを3pt、線種を破線にするというコードであれば以下のようになります

```
proc sgplot data = lib1.nhefs;  
  scatter x=sbp y=dbp / markerattrs=(color=black size=5pt symbol=squarefilled);  
  reg x=sbp y=dbp / nomarkers lineattrs=(color=red thickness=3pt pattern=dash);  
run;
```



プロットのオプションで `group` を使い、グループごとに描画している場合について、グループごとに属性を指定する方法は複数ありますが、簡便な `styleattrs` について紹介します


```
ods graphics / attrpriority=none;
```

```

proc sgplot data = lib1.nhefs;

styleattrs
datacontrastcolors=(red blue)
datasymbols=(triangle circle)
datalinepatterns=(solid dash)
;
scatter x=sbp y=dbp /group= sex;
reg x=sbp y=dbp / group= sex  nomarkers lineattrs=(thickness=2pt);
run;

```



styleattrs から ;の間に、変えたい属性について、リスト形式で指定します。

上記の場合、一つ目のグループは、赤色で、三角形で、直線、2つ目は青色で、丸形で、破線となります。


ods graphics / attrpriority=none;の部分について、デフォルトである attrpriority=color だと、色×シンボルの種類分、グループ属性を作成してしまうので、グループ数が膨大な場合を除くと none を指定したほうがわかりやすく、今回はそのようにしています。

3.2 軸と凡例の体裁

変数の値をそのままプロットに示すのではなく 0 であれば男性と表示したいなどの場合、SAS では format プロシジャで出力用のフォーマットを作成し、それをプロットで指定しま

す。また y 軸の設定については yaxis ステートメント, x 軸の設定については xaxis ステートメント, 凡例の設定については keylegend を指定します。

```
proc format;
  value sexf 0="男性" 1="女性";
run;
proc sgplot data = lib1.nhefs;
  scatter x=sbp y=dbp / group = sex;
  format sex sexf.;
  yaxis label="拡張期血圧" values=(30 to 150 by 20);
  xaxis label="収縮期血圧" values=(80 100 150 180 200);
  keylegend /title="性別" location = inside position=topleft across=1;
run;
```



4. 参考

SAS で解析あるいはグラフを作成する場合において、データの前処理が必要になるケースが多いです。今回は可視化部分にフィーチャーする都合上、データセットからダイレクトにグラフを作成しています。データステップの基本については以下を参照下さい

[ウェブサイト]SAS 忘備録：SAS 入門

<https://sas->

[boubi.blogspot.com/search/label/%5B%E2%96%BC%20SAS%E5%85%A5%E9%96%80](https://sas-boubi.blogspot.com/search/label/%5B%E2%96%BC%20SAS%E5%85%A5%E9%96%80)

[書籍]統計解析ソフト SAS

<https://www.cutt.co.jp/book/978-4-87783-503-3.html>

グラフ描画においては、基本的にステートメントやオプションが膨大にあるため SAS 社公式のウェブリファレンスの最新を検索して、リファレンスを見ながらコーディングすることが多いです。

[SAS 社公式ウェブリファレンス]

SAS® 9.4 and SAS® Viya® 3.5 Programming Documentation

https://documentation.sas.com/doc/en/pgmsascdc/9.4_3.5/grstatproc/n0yjdd910dh59zn1t0odgupaj4v9.htm

作成者：森岡 裕(イーピーエス株式会社)

Mail: morioka.yutaka038@eps.co.jp